## (DRAFT)

# Notes on Formulating and Solving Computable General Equilibrium Models within GAMS 

by

Dhazn Gillig<br>Research Scientist<br>Department of Agricultural Economics<br>College Station, TX<br>dgillig@taexgw.tamu.edu<br>(979) 845-3153<br>\section*{Bruce A. McCarl}<br>Professor<br>Department of Agricultural Economics<br>College Station, TX<br>mccarl@tamu.edu<br>(979) 845-1706

July 2002

## Notes on Solving Computable General Equilibrium Models in GAMS

GAMS is commonly used to solve computable general equilibrium (CGE) models. This document presents background material on CGE modeling with GAMS first covering the general structure of a CGE and then focusing on the GAMS formulation and subsequent solution of the problem. Note in this coverage we will not be using the MPSGE approach as we prefer the more structural approach. Readers wishing details on MPSGE should consult the materials developed by Rutherford on http://debreu.colorado.edu/mainpage/mpsge.htm

The majority of our discussion will be presented in the context of a two goods, two household, two factor setting in a closed economy. This setting is adapted from an example provided in GAMS model library (TWO3MCP) which is based on the work of Shoven and Whalley (1984).

## CGE Model Background

CGE models depict entire economies. In such models prices and production/ consumption in both factor and output markets are endogenous as is income. At equilibrium there are several characteristics of CGE model solutions:

1. The total market demand equals the total market supply for each and every factor and output market.
2. Prices are set so that equilibrium profits of firms are zero with all rents accruing to factors.
3. Household incomes equal household expenditures.
4. Government tax revenues equal government expenditures including subsidy payments.

## Basic Notation

From here on we assume a perfectly competitive economy that produces a number of goods denoted by $j$ which contains a number of households denoted by $h$ using the basic factor inputs labor and capital. The production quantities, factor usages, and associated prices are defined as follows:
$Q_{j}$ is the production of goods by the $j^{\text {th }}$ sector, $P_{j}$ is the price of goods produced by the $j^{\text {th }}$ sector, $L_{j}$ is the usage of labor by the $j^{\text {th }}$ sector, $W_{L}$ is the price of labor (the wage rate),
$a_{j 1, j}$ is the amount of goods used from sector $j 1$ when producing one unit of goods in the $j^{\text {th }}$ sector,
$K_{j}$ is the usage of capital in the $j^{\text {th }}$ sector, $W_{K}$ is the price of capital, and
$X_{j h}$ is the consumption of the goods produced by sector $j$ by household $h$.

We also assume that the factors are all owned by households where
$\bar{L}_{h} \quad$ is the endowment of labor in household $h$, and
$\bar{K}_{h} \quad$ is the endowment of capital in household $h$.

## Equilibrium Conditions (without Taxes)

A set of non-zero prices $P_{j}, W_{L}$, and $W_{K}$, consumption levels $X_{j h}$, production levels $Q_{j}$ and factor usages $L_{j}$, and $K_{j}$ constitutes an economic equilibrium solution (also called a Walrasian equilibrium) and a solution to a CGE of the situation if
I. The total demand is less than or equal to the total supply in every factor market. The total supply is the sum across the household endowments; in other words, the excess demand in the factor input markets is less than or equal to zero:

$$
\begin{align*}
& \sum_{j} L_{j}-\sum_{h} \bar{L}_{h} \leq 0  \tag{1}\\
& \sum_{j} K_{j}-\sum_{h} \bar{K}_{h} \leq 0 . \tag{2}
\end{align*}
$$

II. The total demand in every output market including consumer and intermediate production usage is less than or equal to the total supply in that market. In other words, the excess demand in each output market is less than or equal to zero:

$$
\begin{equation*}
\sum_{h} X_{j h}+\sum_{j 1} \mathrm{a}_{\mathrm{j}, \mathrm{j} 1} Q_{j 1}-Q_{j} \leq 0 \quad \forall j \tag{3}
\end{equation*}
$$

III. Profits are zero in each sector:

$$
\begin{equation*}
P_{j} Q_{j}=\sum_{j 1} P_{j 1} \mathbf{a}_{\mathbf{j} 1, \mathbf{j}} Q_{j}+W_{L} L_{j}+W_{K} K_{j} \quad \forall j \tag{4}
\end{equation*}
$$

Walras' Law also introduces a requirement. Namely,
"Walras' Law: For any price vector $\boldsymbol{P}$, we have $\mathbf{P Z}(\boldsymbol{P})=0$; i.e., the value of the excess demand is identically zero." (Varian, page 317)
where $\mathbf{Z}(\mathbf{p})$ is the excess demand in the factor and product markets as shown in equations (1), (2). and (3). Then we get the additional requirement that

$$
\begin{equation*}
\sum_{j} P_{j}\left(Q_{j}-\sum_{j 1} \mathrm{a}_{\mathrm{j}, \mathrm{j} 1} Q_{j 1}-\sum_{h} X_{j h}\right)+W_{L}\left(\sum_{j} L_{j}-\sum_{h} \bar{L}_{h}\right)+W_{K}\left(\sum_{j} K_{j}-\sum_{h} \bar{K}_{h}\right)=0 . \tag{5}
\end{equation*}
$$

Walras' Law and equation (5) coupled with an assumption of nonnegative prices and excess demands also imply a complementarity between prices and excess demand (Ferris and Munson, Ferris and Pang, and Manne). Namely, if the total demand is less than the total supply for the commodity $j$ market then the price in that market must be zero. Conversely, prices will be nonzero only if supply equals demand (Varian). For example, for the factors the factor prices must be zero and the factors not all used up or for non zero prices to exist the factors must be all consumed:

$$
\begin{equation*}
0 \leq W_{L} \perp \sum_{j} L_{j}-\sum_{h} \bar{L}_{h} \leq 0 \tag{6}
\end{equation*}
$$

$$
\begin{equation*}
0 \leq W_{K} \perp \sum_{j} K_{j}-\sum_{h} \bar{K}_{h} \leq 0 \tag{7}
\end{equation*}
$$

Similarly the product prices must be zero and the products not all consumed or for non-zero prices to exist the products must be all consumed:

$$
\begin{equation*}
0 \leq P_{j} \perp \sum_{h} X_{j h}+\sum_{j 1} \mathrm{a}_{\mathrm{j}, \mathrm{j} 1} Q_{j 1}-Q_{j} \leq 0 \tag{8}
\end{equation*}
$$

For the firm, profits must equal zero and a non-zero production level achieved or profits can be less than costs without the firm producing.

$$
\begin{equation*}
0 \leq Q_{j} \perp P_{j} Q_{j} \leq \sum_{j 1} \mathrm{P}_{\mathrm{j} 1} \mathrm{a}_{\mathrm{j} 1, \mathrm{j}} Q_{j}+W_{L} L_{j}+W_{K} K_{j} \tag{9}
\end{equation*}
$$

Finally household income is defined by

$$
\begin{equation*}
0 \leq \text { Income }_{h} \perp \text { Income }_{h} \geq W_{L} \bar{L}_{h}+W_{K} \bar{K}_{h} \tag{10}
\end{equation*}
$$

where the $\perp$ notation means that at least one of the adjacent inequalities must be satisfied as an equality. These complementary relationships are important elements of the solution method employed. Namely, we use the solver PATH to impose the complementarity (Ferris and Munson, Ferris and Pang).

## Determining Consumer Demand

Household goods consumption is determined by assumptions about consumer behavior. Consumers are generally assumed to maximize utility subject to a budget constraint. The assumed form of the utility is a constant elasticity of substitution (CES) function. Note other functional forms such as Cobb Douglas, Linear Expenditure System, or Translog can also be used. Under the assumed functional form we may now express household demand for goods by deriving household level product demand relations. The CES utility function $\left[U_{h}(\cdot)\right.$ ] is assumed strictly quasi-concave and differentiable and is given by:

$$
U_{h}=\left[\sum_{j}\left(\alpha_{j h}\right)^{1 / \sigma_{h}}\left(X_{j h}\right)^{\left(\sigma_{h}-1\right) / \sigma_{h}}\right]^{\sigma_{h} /\left(\sigma_{h}-1\right)}
$$

where
$\alpha_{j h}$ is the expenditure share for consumption spending on goods $j$ for household $h$, and $\sigma_{h}$ is the elasticity of substitution in the household $h$ function.

Assume each consumer operates subject to a budget constraint and has an initial endowment of factors, but no initial endowments of goods. The budget constraint is then

$$
\sum_{j} P_{j} X_{j h} \leq W_{L} \bar{L}_{h}+W_{K} \bar{K}_{h}=\text { Income }_{h}
$$

where Income $_{h}$ is household $h$ income. In turn, consumer demand functions can be derived. Such functions will be homogenous of degree zero in prices implying that a uniform change in all prices will not affect quantities demanded and in turn the equilibrium. This property also requires prices to be normalized as will be discussed later so we have a unique solution. The resultant demand function for commodity $j$ is:

$$
X_{j h}=\frac{\alpha_{j h}\left(\text { Income }_{h}\right)}{P_{j}^{\sigma_{h}} \sum_{j}\left(\alpha_{j h}\left(P_{j}\right)^{1-\sigma_{h}}\right)} .
$$

In turn, we add an equation to the CGE model of the form

$$
\begin{equation*}
0 \leq X_{j h} \perp X_{j h}-\frac{\alpha_{j h}\left(\text { Income }_{h}\right)}{P_{j}^{\sigma_{h}} \sum_{j}\left(\alpha_{j h}\left(P_{j}\right)^{1-\sigma_{h}}\right)} \leq 0 \forall j \& h \tag{11}
\end{equation*}
$$

## Producer Factor Demand

The quantity of factors used in production is determined by assumptions about producer behavior. CGE models are typically built under the assumption that producers are maximizing profits with production characterized by a known production function under constant returns to
scale and a constant elasticity of substitution (CES) function. Also, note that other functional forms such as fixed coefficients or Cobb Douglas can also be used. The CES production function is given by:

$$
\left.Q_{j}=\phi_{j} \mid \delta_{j} L_{j}{ }^{\left(\sigma_{j}-1\right) / \sigma_{j}}+\left(1-\delta_{j}\right) K_{j}^{\left(\sigma_{j}-1\right) / \sigma_{j}}\right]^{\sigma_{j} /\left(\sigma_{j}-1\right)}
$$

where
$\phi_{j}$ is the scale parameter for sector $j$,
$\delta_{j}$ is another CES function parameter for sector $j$,
$\sigma_{j}$ is the elasticity of substitution of labor and capital in sector $j$
Applying the assumption of cost minimization and given that sector $j$ produces quantity $Q_{j}$ then an expression for factor demand can be derived which is:

$$
\begin{aligned}
& L_{j}=\frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}+\left(1-\delta_{j}\right)\left(\frac{\delta_{j} W_{K}}{\left(1-\delta_{j}\right) W_{L}}\right)^{\left(1-\sigma_{j}\right)}\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)} \\
& K_{j}=\frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right) W_{L}}{\delta_{j} W_{K}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}
\end{aligned}
$$

where $W_{L}$ and $W_{K}$ are the labor $(L)$ and capital $(K)$ factor prices.
These are included in the CGE model as follows

$$
\begin{align*}
& 0 \leq L_{j} \perp \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}+\left(1-\delta_{j}\right)\left(\frac{\delta_{j} W_{K}}{\left(1-\delta_{j}\right) W_{L}}\right)^{\left(1-\sigma_{j}\right)}\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}-L_{j} \leq 0 \text { for all } j  \tag{12}\\
& 0 \leq K_{j} \perp \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right) W_{L}}{\delta_{j} W_{K}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}-K_{j} \leq 0 \text { for all } j \tag{13}
\end{align*}
$$

## Incorporation of taxes

Taxes are a fact of life and a CGE should contain tax provisions. Tax treatments range through a number of possibilities (Shoven and Whalley and Creedy). To include taxes, a
modification is needed in the zero profit condition to include tax payments as a cost of doing business. We also need to model a government disposition of tax revenues. We introduce taxes into the CGE assuming that there are taxes on production net income and taxes on household income. Suppose that the tax rates are
$T_{j} \quad$ percent imposed on corporate net income, and $T_{h} \quad$ percent imposed on household income.

Also suppose that each household type can deduct $F_{h}$. In turn, total tax revenue $(R)$ collected by the government is

$$
R=\sum_{j} T_{j}\left(P_{j} Q_{j}-W_{L} L_{j}-W_{K} K_{j}\right)+\sum_{h} T_{h}\left(W_{L} \bar{L}_{h}+W_{K} \bar{K}_{h}-F_{h}\right),
$$

which applies the corporate tax rate to an accounting of corporate net income and the household income tax to gross revenue less deductions. In turn, we assume the total tax revenue is redistributed to households in the form of transfer payments $\left(T R_{h}\right)$ or is expended on government purchases of goods $\left(G P_{j}\right)$. Assume that government purchases are proportional to the gross revenue at the rate $\left(s_{j}\right)$ as are household transfer payment shares $\left(s_{h}\right)$. Therefore, household $h$ receives $T R_{h}=s_{h} * R$, and the government expenditures on the $j$ th sector are $G P_{j}=s_{j} * R$ where $\sum_{h} s_{h}+\sum_{j} s_{j}=1$, and in turn $\sum_{h} T R_{h}+\sum_{j} G P_{j}=R$.

This introduces modifications to the model above as follows. First, the market balance expands to include government purchases transformed to be in the quantity unit

$$
\begin{equation*}
\sum_{h} X_{j h}+\sum_{j 1} \mathrm{a}_{\mathrm{j}, \mathrm{j} 1} Q_{j 1}-Q_{j}+s_{j} R / P_{j} \leq 0 \tag{3a}
\end{equation*}
$$

Second, the zero profit condition expands to include tax effects

$$
\begin{align*}
0 \leq Q_{j} \perp P_{j} Q_{j} \leq & \sum_{j 1} \mathbf{P}_{\mathbf{j} 1} \mathbf{a}_{\mathbf{j} 1, \mathbf{j}} Q_{j}+W_{L} L_{j}+W_{K} K_{j} \\
& +T_{j}\left(P_{j} Q_{j}-\sum_{j 1} \mathbf{P}_{\mathbf{j} \mathbf{1}} \mathbf{a}_{\mathbf{j} 1, \mathbf{j}} Q_{j}-W_{L} L_{j}-W_{K} K_{j}\right) \tag{9a}
\end{align*}
$$

Third, the household income equation is adjusted to reflect tax incidence and tax revenue redistribution

$$
\begin{equation*}
0 \leq \text { Income }_{h} \perp\left(1-T_{h}\right) *\left(W_{L} \bar{L}_{h}+W_{K} \bar{K}_{h}-F_{h}\right)+s_{h} \mathbf{R} \leq \text { Income }_{h} \tag{10a}
\end{equation*}
$$

Fourth, we add a government income balance

$$
\begin{equation*}
0 \leq R \perp R \leq \sum_{j} T_{j}\left(P_{j} Q_{j}-\sum_{j 1} \mathbf{P}_{\mathbf{j} 1} \mathbf{a}_{\mathbf{j} 1, \mathbf{j}} Q_{j}-W_{L} L_{j}-W_{K} K_{j}\right)+\sum_{h} T_{h}\left(W_{L} \bar{L}_{h}+W_{K} \bar{K}_{h}-F_{h}\right) \tag{14}
\end{equation*}
$$

The complementary condition in equation (14) indicates that the government satisfies its budget constraint when total revenue is positive.

## Casting the model into GAMS

Now suppose we turn our attention to GAMS. Here we will set up a model using the basic parameters of a model by Shoven and Whalley (1984) as implemented in the TWO3MCP model in the GAMS model library. That structure depicts an economy with two factors of production (labor and capital), two commodities are produced (food and nonfood) by sectors with those names; and there are two household classes (farm and nonfarm). We assume familiarity with basic GAMS instructions so will introduce the model in GAMS format.

First, we need to define the basic subscripts or sets. We define three sets. The set
Factors defines the factors of production in the model encompassing labor and capital. The set Sector depicts the $j$ subscript representing the production sectors in the model. The set

Households denotes the set of different household consumption units in the $h$ subscript above.
Then we define the variables. In this case there are 7 as listed below

## Symbols Used for Problem Variables

| Name in <br> Theoretical <br> Model | Corresponding GAMS Item | Brief Description |
| :--- | :--- | :--- |
| $W_{L} W_{K}$ | FACTORPRICE(Factor) | Prices for the factors labor and capital |
| $\mathrm{L}_{\mathrm{j}}, \mathrm{K}_{\mathrm{j}}$ | FACTORQUAN(Factor,Sector) | Factor use by a producing sector |
| $P_{j}$ | COMPRICE(Sector) | Prices of commodities |
| $X_{j h}$ | DemCommod(Households,Sector) | Commodity demand by households |
| $Q_{j}$ | PRODUCTION(Sector) | Production quantity |
| Income $e_{h}$ | HHIncome(households) | Household income |
| $R$ | TAXREVENUE | Total government tax revenue |

We also need GAMS equations associated with each of the equations developed above and these are:

## Symbols Used for Problem Equations

| Eqn Number <br> in Model | Corresponding GAMS Item | Brief Description |
| :--- | :--- | :--- |
| 1,2 | FactorMkt(Factor) | Factor market balances |
| 12,13 | FactorDem(Factor,Sector) | Commodity demand by households |
| 3 a | CommodMkt(Sector) | Commodity market balance |
| 11 | CommodDem(Households,Sector) | Commodity demand by households |
| 9 a | Profit(Sector) | Zero profit condition |
| 10 a | Income(households) | Household budget constraint |
| 14 | GovBal | Government budget constraint |

To specify this model we also need to define a number of data items. These include

## Symbols Used for Problem Data

| Symbol Name <br> Model | Corresponding GAMS Item | Brief Description |
| :---: | :--- | :--- |
| $\sigma_{h}$ | SigmaC(Households) | Elasticity of substitution in household CES |
| $\alpha_{j h}$ | Alpha(Sector,Households) | Consumption share in household CES |


| $\bar{L}_{h}, \bar{K}_{h}$ | Endowment(Factor,Households) | Household endowments of factors |
| :--- | :--- | :--- |
| $\mathrm{a}_{j 1, j}$ | Intermediateuse(Sector1,Sector) | Use of goods in sector1 when producing in sector |
| $\phi_{j}$ | Phi(Sector) | Scale parameter in CES production function |
| $\delta_{j}$ | Delta(Factor,Sector) | Distribution parameter in CES production |
| $\sigma_{j}$ | Sigma(Sector) | Elasticity of production factor substitution |
| $\mathrm{s}_{\mathrm{h}}$ | TaxShare(Households) | Household share of tax disbursements |
| $\mathrm{s}_{\mathrm{j}}$ | Governmentpurch(Sector) | Government goods purchase dependence on <br> Revenue |
| $\mathrm{T}_{\mathrm{h}}$ | Incometax | Household tax level |
| $\mathrm{T}_{\mathrm{j}}$ | CorporateTax(Sector) | Tax on corporate earnings |
| $\mathrm{F}_{\mathrm{h}}$ | TaxExemption(Households) | Household tax exemptions |

The class of models that this CGE falls into is the so called mixed complementarity class
(an MCP) in GAMS. To solve such models we must have a model with

1) complementarity requirements associated with each equation since each has the requirement denoted by $\perp$,
2) a consistent set of variable and equation definitions where each variable is of the same dimension as an associated equation that it is complementary with,
3) a model where no variable is complementary with more than one equation or vice versa, and
4) a model where every variable and equation has a complementary partner.

Our model satisfies these requirements and the complementary relationships are expressed in the model statement. Specifically, in GAMS we express this in the MODEL statement employing the notation involving the periods in the statement below

MODEL CGEModel
/FactorMkt.FACTORPRICE, FactorDem.FACTORQUAN, CommodMkt.COMPRICE,commoddem.DemCommod

This notation shows the complementary partners and indicates that the complementary relations in the table below need to be satisfied in the solution.

## Complementary Items

| Variable Name | Equation Name |
| :--- | :--- |
| FACTORPRICE(Factor) | FactorMkt(Factor) |
| FACTORQUAN(Factor,Sector) | FactorDem(Factor,Sector) |
| COMPRICE(Sector) | CommodMkt(Sector) |
| DemCommod(Households,Sector) | CommodDem(Households,Sector) |
| PRODUCTION(Sector) | Profit(Sector) |
| HHIncome(households) | Income(households) |
| GovtIncome | GovBal |

Note how the dimensions of the items are the same.
The variables mentioned above are declared in GAMS using the statement below

## POSITIVE VARIABLES

FACTORPRICE(Factor) Prices for factors
FACTORQUAN(Factor,Sector)
COMPRICE(Sector)
DemCommod(Households,Sector)
PRODUCTION(Sector)
HHIncome(households)
GovtIncome
Factor use by a sector
Prices of commodities
Demand by households
Production quantity
Household income
Government tax revenue

The equations are defined as follows

## EQUATIONS

FactorMkt(Factor)
FactorDem(Factor,Sector)
CommodMkt(Sector)
CommodDem(Households,Sector)
Profit(Sector)
Income(households)
GovBal

Factor market balances
Factor demand by a sector
Commodity market balance
Commodity demand by households
Zero profit condition
Household budget constraint
Government budget constraint ;
and are specified as follows

## Factor Market Balance

For factors, the factor used in production is less than or equal to the total supply in every factor input market which is composed from the household endowments.

FactorMkt(Factor)..
sum(households,Endowment(Factor,households)) =G=
sum(Sector,FactorQuan(Factor,Sector)) ;

## Commodity Market Balance

For commodities, the total demand in the output market for each sector including consumer demand, intermediate production usage, and government purchases are less than or equal to the total production in that market.

CommodMkt(Sector)..
Production(Sector) =G= sum(households,DemCommod(Households,Sector))
+Governmentpurch(sector)*GovtIncome/COMPRICE(Sector)
+sum(othersector, intermediateuse(sector,othersector)*production(othersector));

## Firm Zero Profit Condition

For each production sector, revenues are less than or equal to costs with in effect all rents allocated to factors. Thus the total revenue is less than factor usage costs plus costs of intermediate products obtained plus tax payments.

```
Profit(Sector)..
    sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))
    +sum(othersector,
    ComPrice(otherSector)*intermediateuse(othersector,sector)*production(sector))
+corporatetax(sector)*(
    ComPrice(Sector)* Production(Sector)
    -sum(othersector,
```

```
    ComPrice(OtherSector)*intermediateuse(Othersector,sector)*production(sector))
    -sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))
=G= ComPrice(Sector)* Production(Sector) ;
```


## Household Income

For each household, the income is less than or equal to the household income from factor sales adjusted for taxes and tax deductions plus transfer payments from the government.

```
HHIncome(households) =G=
    (1-incometax)*sum(Factor,Endowment(Factor,households) * FactorPrice(Factor))
    + incometax * TaxDeduction(Households)
    + TaxShare(households) *TaxRevenue ;
```


## Government Tax Income

For the government tax, the government income is less than or equal to taxes from households applied to factor income less deductions plus corporate taxes applied to net income.

GovBal..

```
GovtIncome =G=
        sum(sector,corporatetax(sector)*
        ( ComPrice(Sector)* Production(Sector)
        -sum(OtherSector,
    ComPrice(OtherSector)*intermediateuse(othersector,sector)
    *production(sector1))
        -sum(Factor,FactorPrice(Factor)*FactorQuan(Factor,Sector))))
        +sum(households,incometax
        *(sum(Factor,Endowment(Factor,households) * FactorPrice(Factor))
        -TaxDeduction(Households)));
```


## Household Commodity Demand

For each household we assume a CES utility function and derive commodity demand subject to a budget constraint. In turn, this lets us develop a formula for commodity demand that we imbed in the model. Demand is a function of income and prices. The function is also homogeneous of degree zero in prices which will require other actions later.

```
CommodDem(Households,Sector)..
DemCommod(Households,Sector) \(=\mathrm{E}=\)
    (HHIncome(households)/sum(Sector1,alpha(Sector1,households)
    *ComPrice(Sector1)**(1-SigmaC(households)))
    )*Alpha(Sector,households) * (1
/ComPrice(Sector))**sigmaC(households);
```


## Producer Factor Demand

For each producer we assume a CES production function and derive factor demand assuming cost minimizing production of a given quantity of product. In turn, this lets us develop a formula for factor demand that we imbed in the model. Demand is a function of production quantity and factor prices.

```
FactorDem(Factor,Sector)..
    FACTORQUAN(Factor,Sector) =g=
    Production(Sector)*Phi(Sector) **(sigma(Sector)-1)
        *(Delta(Factor,Sector)
            *( sum(Factor1,Delta(Factor1,Sector) **sigma(Sector)
            *(FactorPrice(Factor1)) **(1 - sigma(Sector)))
        **(1/(1-sigma(Sector)))/Phi(Sector))/
    FactorPrice(Factor))**sigma(Sector) ;
```


## Complementarity

Walras's Law introduces a requirement. Namely, for any price vector P , we have $\mathrm{PZ}(\mathrm{P})=0$; i.e., the value of the excess demand, $\mathrm{Z}(\mathrm{P})$, is identically zero." (Varian, page 317). This implies price or excess demand $=0$; This makes our model complementary in terms of following items

Variable Name<br>FACTORPRICE(Factor)<br>FACTORQUAN(Factor,Sector)<br>COMPRICE(Sector)<br>DemCommod(Households,Sector)<br>PRODUCTION(Sector)<br>HHIncome(Households)<br>TAXREVENUE<br>\section*{Complementary Equation}<br>FactorMkt(Factor)<br>FactorDem(Factor,Sector)<br>CommodMkt(Sector)<br>CommodDem(Households,Sector)<br>Profit(Sector)<br>Income(Households)<br>GovBal

We reflect that requirement in GAMS as follows

MODEL CGEModel /FactorMkt.FACTORPRICE, FactorDem.FACTORQUAN, CommodMkt.COMPRICE, commoddem.DemCommod Profit.PRODUCTION, Income.Hhincome, GovBal.GovtIncome/;

## Normalizing Prices

The formulation needs to overcome the homogeneous of degree zero property since an infinite number of prices can yield a given solution. This is done either by setting the income for one household equal to one or the price of a commodity to one. We choose the former

HHincome.L(households) =
$\operatorname{sum}$ (Factor, FactorPrice.l(Factor) * Endowment(Factor,Households));
HHincome.fx(Households)\$(ord(Households) =1) = HHincome.l(Households);

## Starting Points and Bounds

We also use starting points and lower bounds to avoid numerical problems.

## Solving

This problem in turn is best solved with the PATH solver. Details on that solver can be found at http://www.gams.com/solvers/solvers.htm\#PATH. So we choose PATH as the solver and solve with

OPTION MCP = PATH;
SOLVE CGEModel USING MCP;

## The Solution

In turn, a solution arises and you can use normal report writing, graphics etc. as with any other model

```
---- VAR FACTORPRICE Prices for the factors
```


---- VAR DemCommod Commodity Demand LOWER LEVEL UPPER MARGINAL

| Farm | . Food |  | 3.230 | +INF |
| :---: | :---: | :---: | :---: | :---: |
| Farm | . NonFood |  | 4.299 | +INF |
| NonFarm.Food |  |  | 50.429 | +INF |
| NonFarm. NonFood |  |  | 42.155 | +INF |


| ---- VAR | PRODUCTION | Production Quantity |  |  |
| :--- | :---: | :--- | :---: | :---: |
|  | LOWER | LEVEL | UPPER | MARGINAL |
| Food | . | 53.659 | + INF | . |
| NonFood | . | 46.454 | + INF | . |

---- VAR HHIncome Household income

|  | LOWER | LEVEL | UPPER | MARGINAL |
| :--- | :---: | :---: | :---: | :---: |
| Farm | 7.000 | 7.000 | 7.000 | $9.429 E-12$ |
| NonFarm | 0.010 | 84.895 | +INF | • |
|  |  |  |  |  |
|  |  | LOWER | LEVEL | UPPER |
|  |  |  |  |  |
| MARGINAL |  | 0.092 | +INF |  |

## Calibration

In previous sections we show how to develop a CGE model and implement it in GAMS to solve for the equilibrium solutions of price ( $P_{j}, W_{L}$, and $W_{K}$ ), consumption levels ( $X_{j h}$ ), production levels $\left(Q_{j}\right)$ and factor usages $\left(L_{j}\right.$ and $\left.K_{j}\right)$ given specific functional forms along with their parameter values. However, this forms a fairly difficult nonlinear system of equations and one must first insure that a solution exists and that the data within the model are consistent. This has come to be known as the calibration or a so-called "benchmark equilibrium" problem in CGE modeling. (Shoven and Whalley, 1992, or Robinson, and El-Said, 2000).

The first step in the calibration problem involves making certain that the benchmark equilibrium data is consistent with the equilibrium conditions for a base year of data. In practice, the benchmark equilibrium data directly obtained from national accounts (input-output tables) or other government agencies are unlikely to be consistent with the general equilibrium conditions (i.e. the producer factor cost spent on labor or capital may not equal the household labor or capital income). As a result, adjustments are needed to ensure that the equilibrium conditions hold. Numerous adjustments can be applied on a case by case basis. Section v in Shoven and Whalley (1992) discusses procedures for constructing and adjusting the benchmark equilibrium data set.

After constructing the benchmark equilibrium data, the next step is to determine parameter values conforming to this benchmark equilibrium data. Because the benchmark equilibrium data mainly provide information on price and quantity, in some cases such additional extraneous data are required. For example, recall the CES production function previously discussed,

$$
Q_{j}=\phi_{j}\left[\delta_{j} L_{j}^{\left(\sigma_{j}-1\right) / \sigma_{j}}+\left(1-\delta_{j}\right) K_{j}^{\left(\sigma_{j}-1\right) / \sigma_{j}}\right]^{\sigma_{j} /\left(\sigma_{j}-1\right)},
$$

this CES production function has 3 parameters, namely $\phi_{j}, \delta_{j}$, and $\sigma_{j}$; however, the benchmark equilibrium data typically provide only information on the ratios of factor price and factor demand, but do not provide information on the elasticity of substitution. Hence, the value of elasticity of substitution must be exogenously specified. The value of elasticity of substitution can be obtained either from the econometric estimation or from the literature. After the elasticity of substitution is available, the parameters ( $\phi_{j}$, and $\delta_{j}$ ) associated with the CES production function can be calculated as shown below.

From the factor demands derived previously,

$$
L_{j}=\frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}+\left(1-\delta_{j}\right)\left(\frac{\delta_{j} W_{K}}{\left(1-\delta_{j}\right) W_{L}}\right)^{\left(1-\sigma_{j}\right)}\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}
$$

and

$$
K_{j}=\frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right) W_{L}}{\delta_{j} W_{K}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}
$$

we can arrange these into

$$
\begin{align*}
& W_{L} L_{j}=W_{L} \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}+\left(1-\delta_{j}\right)\left(\frac{\delta_{j} W_{K}}{\left(1-\delta_{j}\right) W_{L}}\right)^{\left(1-\sigma_{j}\right)}\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}  \tag{15}\\
& W_{K} K_{j}=W_{K} \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right) W_{L}}{\delta_{j} W_{K}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)} \tag{16}
\end{align*}
$$

Then factor shares for sector $j, \omega_{K L}^{j}$, and $\omega_{L K}^{j}$ is equal to
$\square(16) \div(15), \quad \omega_{L K}^{j}=\frac{L_{j} W_{L}}{K_{j} W_{K}}=\frac{\delta_{j}^{\sigma} W_{K}^{1-\sigma}}{\left(1-\delta_{j}\right)^{\sigma} W_{L}^{1-\sigma}}$,
and

$$
\begin{equation*}
\omega_{K L}^{j}=\frac{K_{j} W_{K}}{L_{j} W_{L}}=\frac{\left(1-\delta_{j}\right)^{\sigma} W_{L}^{1-\sigma}}{\delta_{j}^{\sigma} W_{K}^{1-\sigma}} \tag{17b}
\end{equation*}
$$

With the assumption that the benchmark equilibrium factor prices $W_{L}$ and $W_{K}=1$, then we can solve for $\delta_{j}$ from equation (17a) expressed in term of $\omega_{L K}^{j}$,

$$
\delta_{j}=\frac{\left(\omega_{L K}^{j}\right)^{1 / \sigma}}{1+\left(\omega_{L K}^{j}\right)^{1 / \sigma}},
$$

where $\omega_{L K}^{j}=L_{j} / K_{j}$ and is directly provided from the benchmark equilibrium. Note that by solving equation (17b), $\delta_{j}$ can also be expressed in term of $\omega_{K L}^{j}$,

$$
\delta_{j}=\frac{1}{1+\left(\omega_{K L}^{j}\right)^{1 / \sigma}}
$$

where $\omega_{K L}^{j}=K_{j} / L_{j}$ and is also directly provided from the benchmark equilibrium. Next, the value of the sector $j$ scale parameter, $\phi_{j}$, can be determined through the zero profit conditions:

$$
\begin{aligned}
P_{j} Q_{j}= & W_{L} \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right)}{\delta_{j}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)} \\
& +W_{K} \frac{1}{\phi_{j}} Q_{j}\left[\delta_{j}\left(\frac{\left(1-\delta_{j}\right)}{\delta_{j}}\right)^{\left(1-\sigma_{j}\right)}+\left(1-\delta_{j}\right)\right]^{\sigma_{j} /\left(1-\sigma_{j}\right)}
\end{aligned}
$$

With the assumption that the benchmark equilibrium market prices $\left(P_{j}\right)$, and factor prices $\left(W_{L}\right.$, and $W_{K}$ ) are equal to 1 , then we can solve for $\phi_{j}$

$$
\phi_{j}=\left(\left(\delta_{j}\right)^{\sigma_{j}}+\left(1-\delta_{j}\right)^{\sigma_{j}}\right)^{1 /\left(1-\sigma_{j}\right)} .
$$

The determination of the parameter values in the consumer side follows a similar procedure. Recall, the consumer utility function,

$$
U_{h}=\sum_{j}\left[\left(\alpha_{j h}\right)^{1 / \sigma_{h}}\left(X_{j h}\right)^{\left(\sigma_{h}-1\right) / \sigma_{h}}\right]^{\sigma_{h} /\left(\sigma_{h}-1\right)}
$$

but only the expenditure share for consumption spending on goods $j$ for household $h, \alpha_{j h}$, is calculated,

$$
\alpha_{j h}=\left(\frac{\left(\omega_{h}^{j}\right)}{\left(\left(\theta_{h}^{i j^{\prime}}\right)^{1-\sigma_{h}}+\omega_{h}^{j}\right)}\right)
$$

where $\omega_{h}^{j}$ is the ratio of the household share of total expenditure on goods $j$ and $j^{\prime}$,

$$
\omega_{h}^{j}=P_{j} X_{j h} / P_{j^{\prime}} X_{j^{\prime} h}
$$

and $\theta_{h}^{i j^{\prime}}$ is the price ratio of good $j$ and $j^{\prime}, \theta_{h}^{i j^{\prime}}=P_{j} / P_{j^{\prime}}$. The elasticity of substitution in the household, $\sigma_{h}$, is generally obtained from the literature which uses the own-price elasticity as an approximation as shown in Shoven and Whalley (1992).

These parameter values once generated can be used in further model applications to solve for alternative equilibria which result from policy changes.

## Full GAMS Model



```
*( sum(Factor1,Delta(Factor1,Sector)**sigma(Sector)
                    *(FactorPrice(Factor1))**(1 - sigma(Sector)))
**(1/(1-sigma(Sector)))/Phi(Sector))/
    FactorPrice(Factor))**sigma(Sector) ;
```

CommodDem(Households, Sector).
DemCommod(Households, Sector) =E=
(HHIncome (households)
/sum (otherSector, alpha (otherSector, households)
*ComPrice (OtherSector) **(1-SigmaC (households)))
)*Alpha(Sector,households) * (1 /ComPrice(Sector))**sigmaC(households);
CommodMkt (Sector)..
Production(Sector) $=G=$
sum (households, DemCommod (Households, Sector))
+ Governmentpurch(sector)*TaxRevenue/COMPRICE (Sector)
+ sum(othersector,intermediateuse(sector,othersector)*production(othersector));
Profit(Sector)..
sum (Factor, FactorPrice (Factor) *FactorQuan (Factor, Sector))
+corporatetax(sector)*(
ComPrice(Sector)* Production(Sector)
-sum (othersector, ComPrice (OtherSector) *intermediateuse (othersector, sector)
*production(sector))
-sum (Factor, FactorPrice (Factor) *FactorQuan (Factor, Sector)))
=G= ComPrice(Sector)* Production(Sector) ;
Income(households)..
HHIncome (households) =G=
(1-incometax)*sum(Factor, Endowment (Factor, households) * FactorPrice(Factor))
+ incometax * TaxDeduction (Households)
+ TaxShare(households) *TaxRevenue ;
GovBal..
TaxRevenue $=\mathrm{G}=$
sum (sector, corporatetax (sector) *(
ComPrice(Sector)* Production(Sector)
-sum (Othersector, ComPrice (OtherSector) *intermediateuse (Othersector, sector)
*production(sector))
-sum(Factor, FactorPrice(Factor) *FactorQuan(Factor, Sector))))
+sum (households,incometax
* (sum (Factor, Endowment (Factor, households) * FactorPrice(Factor))
-TaxDeduction(Households)));
MODEL CGEModel /FactorMkt.FACTORPRICE, FactorDem.FACTORQUAN,
CommodMkt. COMPRICE, commoddem. DemCommod
Profit.PRODUCTION, Income.Hhincome, GovBal.TAXREVENUE/;

* lower bounds
FACTORPRICE.LO(Factor) $=0.01$;
COMPRICE.LO (Sector) $=0.01$;
HHincome.Lo(households) $=0.01$;
*starting point
FACTorPRICE.L(Factor) = 1 ;
FACTorQuan.L(Factor, sector) = 1 ;
DemCommod.l (Households, Sector) =1;
COMPRICE.L(Sector) = 1 ;
PRODUCTION.L(Sector) $=10$;
HHincome.L(households) =
sum (Factor, FACTorPRICE.l(Factor) * Endowment(Factor,households));
HHincome.fx(households) (ord(households) =1) = HHincome.L(households);
OPTION MCP = PATH;
SOLVE CGEModel USING MCP;


## Reference

Creedy, J. General Equilibrium and Welfare. Edward Elgar Publishing Limited, Cheltenham, UK, 1996.

Ferris, M. C. and T. S. Munson. "GAMS/PATH User Guide Version 4.3." http://www.gams.com/solvers/solvers.htm\#PATH, 2000.
Ferris, M. C. and J. S. Pang. "Engineering and Economic Applications of Complementarity Problems." SIAM Review, 39:669-713, 1997.

Manne, A. S. Economic Equilibrium: Model Formulation and Solution. Mathematical Programming Study 23, edited by A. S. Manne, A publication of the Mathematical Programming Society, Amsterdam, the Netherlands, 1985.

Robinson, S. and M. El-Said, GAMS Code for Estimating A Social Accounting Matrix (SAM) Using Cross Entropy (CE) Methods, Trade and Macroeconomics Division Discussion Papers, Number 64, Dec 2000, http://www.ifpri.org/divs/tmd/dp/tmdp64.htm

Shoven, J. B. and J. Whalley. "General-Equilibrium with Taxes: A Computational Procedure and an Existence Proof." The Review of Economic Studies, 40:475-489, 1973.

Shoven, J. B. and J. Whalley. "Applied General-Equilibrium Models of Taxation and International Trade: An Introduction and Survey." J. Economic Literature, 22:10071051, 1984.
Shoven, J. B. and J. Whalley. Applying General Equilibrium. Cambridge University Press, New York, 1998.

Varian, H. R. Microeconomic Analysis. W. W. Norton \& Company, Inc., New York, 1992

